Department of the Interior

U.S. Geological Survey

Two FORTRAN programs for the reduction of atmospheric data collected by aircraft along EDM survey lines

by

Elliot T. Endo[1]
Gene Y. Iwatsubo
Lyn J. Topinka

Open-File Report 85-279

[1]David A. Johnston Cascades Volcano Observatory
5400 MacArthur Blvd.
Vancouver, Washington 98661

1985

Abstract

The Cascades Volcano Observatory had a requirement for a computer program to reduce atmospheric data collected for electronic distance measurement surveys. Existing programs could not be easily adapted for this simple need. The program AVEINDEX.FOR described in this report fills this requirement. AVEINDEX.FOR, written in FORTRAN-77 standards, is modular in structure. The modular structure allows easy program modification. Twelve supporting subroutines and functions for AVEINDEX.FOR and one data entry program are also described. Program comments accompany each program and most subroutines.

TABLE OF CONTENTS

FIGURES

Introduction

Berg, et al. (1978) documented several computer programs for reducing EDM (Electronic Distance Measurement) data collected for the support of the lunar-ranging experiment at Haleakala, Maui, Hawaii. The comprehensive programs, written in FORTRAN-66 standards, were tailored to the devices used by the Hawaii Institute of Geophysics for atmospheric data-collection and for the K & E Rangemaster II EDM instrument used for measurements. Other programs available within the U.S. Geological Survey are similarly written and are not easily modified for our usage. The Cascades Volcano Observatory requires a program that simply calculates the mean refractive index for an EDM survey line along which atmospheric data were collected either by helicopter or fixed wing aircraft and augmented by pressure measurements at benchmarks. The more complicated effects of an inversion layer were not considered; however, an attempt was made to minimize the nonlinear effects of the compressibility of air. To meet our goal two programs were written utilizing FORTRAN-77 standards. With the exception of 3 subroutines, all code is new. One program, FLYENTER.FOR, facilitates the manual data entry of flight data collected using printed output from a data logger. The second program, AVEINDEX.FOR, uses the output of FLYENTER.FOR to return the average refractive index along the flight line as a function of the specific wavelength of light used in the actual measurement. To obtain the mean refractive index, a refractive index for each measurement point (for temperature, pressure, and humidity) is calculated and the mean calculated using standard statistics. Most equations used for calculations came from Bomford's 4th edition of Geodesy (1980), published reports, text books, or manufacturer provided information. A few critical subroutines are from Savage and Prescott's program MAIN33 to maintain some similarity in the technique of reducing atmospheric data. Precision thermistors and hygristors were transducers used for temperature and humididty measurements. The transducer housing for/mounting on aircraft and electronics duplicate a setup that has been used for precise EDM survey's by the U.S.G.S. in the Branch of Tectonophysics. Atmospheric pressure measurements were made with Setra Systems pressure transducers and a Baromec precision digital barometer.

Both programs have been written to be easily understood and modified by almost anyone with modest FORTRAN programming experience. To facilitate debugging, program modification, and understanding, the programs are modular in structure. Rather than program efficiency, ease of understanding the program and ease of modification were the primary goals. Twelve subroutines or functions are referenced by the main program AVEINDEX.FOR. The described FORTRAN code has been compiled and tested on a VAX 11/750 located at the Cascades Volcano Observatory in Vancouver, Washington. Programs were written primarily for interactive processing, but they are easily adapted to batch processing through the

use of VMS command procedures. The following paragraphs describe the individual subroutines or functions that are referenced by the program AVEINDEX.FOR.

Primary assumptions or approximations:

1. The program assumes constant flight velocity, hence constant elevation change per unit time, in calculating the refractive index for each temperature- and humidity-measurement point.

2. Horizontal air-pressure gradients are not considered.

3. A simple expression from an atmospheric sciences text (Fleagle and Businger, 1963) is used for calculating saturation water-vapor pressure for air. At 30 degrees Celsius there is a 4% error and at 20 degrees Celsius a 2% error.

4. When calculating the gravity term in the pressure equation, standard acceleration of gravity is taken to be acceleration of gravity at 45 degrees latitude (CRC Handbook). Change per meter elevation is also taken from the CRC Handbook.

5. Thermistors and hygristors are assumed to be from the same lot number. If other lot numbers are used an appropriate file has to be generated for polynomial coefficients that are used in the equation for converting resistance to temperature or humidity.

6. When calculating pressure, the mean pressure between measurement points is estimated by taking the average of two previous values. This mean pressure is used in the e/P term of Bomford's pressure equation (eq. 1.27). For the test data set, calculated air pressure for the reflector end differed from the measured value by 1.5 millibars (over an elevation difference of about 800 meters).

7. For refractive index calculations, a wavelength of .840 micrometers is assumed. For EDM instruments other than the HP3808a, another subroutine for calculating refractive index must be used. RINDEX, in EDMALL library, was written for the K & E Rangemaster 3. To change the main program, change the call from HPINDEX to RINDEX.

8. Except as noted, calculations assume temperature in degrees Kelvin, air pressure and vapor pressure in millibars, elevation in meters, and the acceleration of gravity in cm/sec/sec.

9. Curvature of travel path (or earth) for the light beam is ignored.

Subroutines or functions required

1) APRESS.FOR - subroutine to calculate atmospheric pressure using equation 1.27 of Bomford (1980).

2) GETFLY.FOR - subroutine to read flight data file and place data in arrays.

3) GRATIO.FOR - function returns acceleration of gravity ratio for given elevation.

4) HPINDEX.FOR - subroutine returns index of refraction for red light, .840 micrometers for HP3808a, using equation 1.9 from Bomford (1980).

5 MEAND.FOR - double precision version of subroutine to calculate mean and standard deviation for values in an array.

6) MEANS.FOR - single precision version of above.

7) RVALUE2.FOR - function to convert millivolt readings to resistance using calibration curve determined for CVO-specific electronics.

8) RAWTEMP.FOR - returns raw temperature given measured resistance in ohms. Degrees Kelvin is used throughout program.

9) RDHYG83.FOR - subroutine to read hygristor constants from a table.

10) RH33NEW.FOR - returns humidity given measured resistance using appropriate coefficients for polynomial equations provided by the manufacturer of hygristors.

11) RINDEX.FOR - returns refractive index for Helium-Neon laser, K & E Rangemaster. Wave length of .6328 micrometers is assumed.

12) TEMPOK.FOR - function corrects thermistor temperature for air speed.

13) VAPOR.FOR - returns water vapor pressure in millibars given relative humidity.

Required Include file:

1) COMMON.FOR is an include file containing common blocks and declarations.

Specific equations used for calculations:

Atmospheric pressure - Equation 1.27 from (Bomford, 1980). Pressure is given by:

$$P_i = P_o \exp\left(\frac{-0.03415H}{(G45/G)\ T_m\ (1+0.378e/(P_{i-1} + P_{i-2})/2)_m} \right)$$

Where,

P_0 = base or sea level pressure in millibars

T = mean temperature in degrees Kelvin

H = elevation in meters

E = vapor pressure in millibars

P_i = atmospheric pressure at the ith measurement point

G45 = the acceleration of gravity, in cm/sec/sec at sea level and at 45° N latitude

G = the acceleration of gravity at elevation H

For a measured line with an elevation difference over 800 meters, the calculated pressure for the end point was within 1.5 millibars of the measured value.

GRATIO.FOR

Returns the ratio of the acceleration of gravity at the EDM elevation (45 degrees North latitude) to the acceleration of gravity for a given elevation. G45, or the acceleration of gravity at 45 degrees N Latitude, is 980.621 cm/sec/sec at sea level. For the acceleration of gravity for a given elevation in meters, 0.0003086 cm/sec/sec per meter is subtracted from G45. Constants are from the Chemical Rubber Handbook of Physical Constants (1969).

HPINDEX.FOR

Returns the refractive index of infrared light with a wavelength of 0.840 micrometers, the wavelength specific to the source used for the HP3808a EDM device. Taken from Bomford (1980), equation 1.9, page 44.

$$N_o = 287.604 + 1.6288/\Lambda^2 + 0.0136/\Lambda^4$$

where Λ = 0.840 μmeters

RVALUE2.FOR

Function converts millivolt readings from the DATEL logger to values of resistance in ohms. The linear relation was determined by a least squares fit to data obtained during calibration of the related electronics. A plot of the calibration data is shown in figure 1.

$$R = \text{Resistance} = 15{,}184.37 - 276{,}614.19(x)$$

where x = reading in millivolts
R = ohms

RAWTEMP.FOR

Returns raw temperature given a resistance in ohms. Uses equations and constants provided by the manufacturer of the thermistors used.

RH33NEW.FOR

Like RAWTEMP.FOR, the subroutine uses a polynomial function provided by the manufacturer for the specific lot of hygristors used by the project.

RINDEX.FOR

Computes the refractive index for the Helium-Neon laser used by the K & E Rangemaster 3. Uses same equation as HPINDEX.FOR with the wavelength as 0.6328 micrometers.

TEMPOK.FOR

Corrects temperatures for frictional effects of airspeed. Reference is Savage and Prescott (1973).

$$T = T_1 - \beta V^2$$

T_1 = temperature measured, degrees Celsius
V = aircraft speed in mph
$\beta = 0.6 \times 10^{-4}$

VAPORP.FOR

Calculates saturation vapor pressure for a given temperature and returns vapor pressure in millibars provided the relative humidity is known. The equation for saturation vapor pressure is taken from a textbook by Fleagle and Businger (1963).

$$\log_{10} es = 9.4051 - 2353/T$$

T = temperature in degrees Kelvin
es = saturation vapor pressure in millibars
r = relative humidity $\equiv \frac{e}{es}$

Main Programs

FLYENTER.FOR

Documentation is contained in comment statements at the beginning of the program and requires no other explanation.

AVEINDEX.FOR

The primary purpose of the program is to reduce atmospheric data collected during EDM surveys to a number that best represents the mean refractive index along the survey line. For surveys with large elevation differences between benchmarks. survey points it is inappropriate to use the simple average of end point refractive indices because of the usual exponential decrease in the density of air with elevation. The program does not solve the problems associated with temperature inversion. The second goal is to provide users with easy to understand code and documentation that would facilitate any future program modification. The following paragraph explains some of the more subtle aspects of the program that are not obvious from the comment statements within the program itself.

Pointed out earlier was the assumption that the aircraft was flying at constant velocity, and hence that the rate of elevation change was constant per unit time. The program also assumes that the first data input corresponds to a measurement just above the EDM instrument and the last data input to a measurement just above the reflector. Thus a measured line is divided into N-1 intervals. Each interval end point differs in elevation by the total elevation difference divided by N-1. Atmospheric pressure calculations are based on the interval change in elevation. A flat earth was assumed, and effects of a curved ray travel-path was ignored since most survey lines are less than 10 km in distance. To use Bomford's (1980) equation 1.27, the measured EDM instrument pressure is used for the first calculation rather than the mean between the end points of that first interval. For each subsequent interval the mean pressure for i-2, and i-1 points are used for calculating atmospheric pressure at the ith point. If Bomford's equation were used explicitly, then one would need to have a pressure measurement for the ith point. The procedure used is an approximation that avoids this specific problem. For pressure calculations the reflector elevation is subtracted from the EDM instrument elevation and the subsequent algebraic sign used to determine if pressure increases or decreases.

Temperature corrections for frictional heating of thermistors are specific to the type of transducer shield used by the Cascade Volcano Observatory. Corrected distances given by the program are mark-to-mark distances.

Listing of main programs and required subroutines

APRESS.FOR

```
      SUBROUTINE APRESS(I, ELEV, MEANT, MPRES, MVAPOR, GRATIO, PRES)
C.....SUBROUTINE CALCULATES AIR PRESSURE FOR A GIVEN ELEVATION CHANGE
C.....EQUATION FROM BOMFORD, 1980, 4TH EDITION OF GEODESY, P. 51,
C.... EQUATION 1.27
C.... VARIABLES USED:
C....  GUNPRES = P0 OR BASE PRESSURE IN MILLIBARS
C....  PRES = CALCULATED AIR PRESSURE FOR NEXT POINT, MILLIBARS
C....  MEANT = MEAN TEMPERATURE BETWEEN P0 AND PN
C....  GRATIO = RATIO OF STANDARD GRAVITY AT 45 DEGREES LATITUDE AND
C....           ACCELERATION OF GRAVITY FOR GIVEN ELEVATION
C....  ELEV = ELEVATION OF NEXT POINT IN METERS
C....         SIGN IS MINUS FOR UP AND PLUS FOR DOWN TO KEEP TRACK
C....         OF CHANGE IN PRESSURE
C....  MVAPOR = MEAN VAPOR PRESSURE BETWEEN P0 AND P1 IN MILLIBARS
C....  MPRES = MEAN PRESURE BETWEEN P0 AND PN IN MILLIBARS
C....          FOR FIRST AND SECOND POINTS THIS IS APPROXIMATELY EQUAL
C....          TP P0, OR THE GUN PRESSURE
      INCLUDE 'COMMON.FOR/LIST'
      REAL*8 EPVAPOR,TELEV
      REAL*4 TESTELE
      EPVAPOR = 1.0 + (0.378 * (MVAPOR /MPRES ))
      TELEV = 0.03415 * ELEV
       TESTELE = ABS(ELEV)
      IF(TESTELE.LT.0.) THEN
         PRES = GUNPRES(1)
      ELSE
         PRES = GUNPRES(1) * DEXP( TELEV / (GRATIO * MEANT *  EPVAPOR))
      ENDIF
      RETURN
      END
```

AVEINDEX.FOR

```
C.....PROGRAM AVEINDEX.FOR
C.....USE TO REDUCE FLIGHT DATA RECORDED ON LONG EDM LINES UTILIZING DUAL
C.....THERMISTOR, SINGLE HYGRISTOR SETUP DEVELOPED BY SAVAGE ET. AL.,
C.....A COMPANION DATA ENTRY PROGRAM, FLYENTER.FOR SHOULD BE USED TO
C.....CREATE INDIVIDUAL LINE DATA FILES. PROGRAM CONVERTS VOLTAGE READINGS
C.....FROM A RECORDER TO RESISTANCE FROM A PREVIOUSLY DETERMINED LINEAR
C.....REALTIONSHIP. THIS PARTICULAR PROGRAM USES CALIBRATION DATA FOR THE
C.....CVO ELECTRONICS. THESE RESISTANCE VALUES ARE THEN CONVERTED TO
C.....TEMPERATURE, DEGREES KELVIN AND HUMIDITY USING POLYNOMIAL FUNCTIONS
C.....PROVIDED BY THE MANUFACTURER. TEMPERATURE READINGS ARE
C.....THEN CORRECTED FOR FRICTIONAL HEATING USING AN EQUATION PUBLISHED
C.....BY SAVAGE AND PRESCOT, 1973. GUN PRESSURE IN MILLIBARS AND AIRSPEED
C.....IN MPH ARE USED FOR CALCULATIONS.
C.....FOR VAPOR PRESSURE CALCULATIONS A SIMPLIFIED EQUATION FROM
C.....FLEAGLE, R.G., AND J.A. BUSINGER, 1963, ATMOSPHERIC PHYSICS, ACADEMIC
C.....PRESS, N.Y., 346 P. WAS USED.
C.....AIR PRESSURE IS CALCULATED USING EQUATION 1.27 FORM BOMFORD'S
C.....FORTH EDITION OF GEODESY. FOR THE E/P TERM MEAN PRESSURE IS
C.....CALCULATED USING 2 EARLIER DATA POINTS.  FOR FIRST CALCULATION
C.....GUN PRESSURE IS USED. FOR THE GRAVITY TERM IN THE PRESSURE
C.....EQUATION, G45 IS TAKEN AS STANDARD GRAVITY AND SUBSEQUENTLY
C.....CORRECTED FOR ELEVATION. THE REFRACTIVE INDEX FOR EACH DATA
C.....POINT MEASURED ON THE FLIGHT LINE IS CALCULATED USING THE ABOVE
C.....INFORMATION, THEN A MEAN OF ALL POINTS IS CALCULATED AND USED
C.....FOR THE CORRECTED DISTANCE. TWO DISTANCES ARE GIVEN, ONE
C.....USES BOMFORD'S EQUATION  1.9 FOR CALCULATING REFRACTIVE INDEX
C.....FOR STP, AND THE OTHER USING AN EQUATION USED BY HP FOR
C.....CALCULATING REFRACTIVE INDEX FOR STP (15 DEGREES CELSIUS).
C.....FILES REQUIRED:
C..... UNIT = 2, HYGRISTOR CONSTANTS
C..... UNIT = 3, FLGHT DATA
C..... UNIT = 7, PRINTER OUTPUT
C....
C....  VARIABLES USED:
C        AMEAN = MEAN, OUTPUT FROM MEANS OR MEAND
C        FILEOUT = OUTPUT FILE NAME WHOSE DEFAULT FILE TYPE  IS .LIS
C        LUN = UNIT NUMBER FOR INPUT DATA
C        DISTANCE = EDM READOUT CORRECTED FOR ENVIROMENTAL CONDITIONS
C        FNAME = INPUT FILE NAME, DEFAULT TYPE IS .DAT
C        MEANT = MEAN TEMPERATURE BETWEEN N - N-1 POINTS, USED TO CALCULATE
C                AIR PRESSURE
C        MPRES, MVAPOR = MEAN PRESSURE, VAPOR PRESSURE, FOR SAME PURPOSE AS
C                        AS MEANT
C        NHYG = LOT NUMBER FOR HYGRISTOR DATA
C        ELEVDIFF = ELEVATION DIFFERENCE BETWEEN GUN AND REFLECTOR IN METERS
C        GELEV = ELEVATION USED FOR ACCELERATION OF GRAVITY CALCULATION
C        STDEV = STANDARD DEVIATION, OUTPUT FROM MEAND OR MEANS
```

```
C        ELEVINT = ELEVATION INTERVAL USED FOR PRESSURE CALCULATIONS
C        GRAV = ACCELERATION OF GRAVITY IN CM/SEC**2
C     ARRAYS USED:
C        AIRSPEED = AIRCRAFT SPEED IN MPH
C        DAT = HYGRISTOR LOT DATE
C        ELEVG = GUN ELEVATION IN METERS
C        ELEVR = REFLECTOR ELEVATION IN METERS
C        EVAPOR = VAPOR PRESSURE IN MILLIBARS
C        GUNPRES = GUN PRESSURE IN MILLIBARS
C        GUNTEMP = GUN TEMPERATURE IN CENTIGRADE
C        HT = HYGRISTOR VOLTMETER READING IN MILLIVOLTS
C        LOT = HYGRISTOR LOT NUMBER
C        NMINUS1 = REFRACTIVE INDEX MINUS 1.0
C        NUMBER = NUMBER OF READINGS, THIS PROGRAM LIMITED TO 50
C        PRES = CALCULATED AIR PRESSURE, IN MILLIBARS
C        READOUT = EDM READOUT IN METERS
C        REFPRES = REFLECTOR ATMOSPHERIC PRESSURE IN MILLIBARS
C        REFTEMP = REFLECTOR TEMPERATURE IN DEGREES CENTIGRADE
C        RESISHT = HYGRISTOR RESISTANCE IN OHMS
C        RESISRT = THEMISTOR RESISTANCE IN OHMS, ASSUMES PARALLEL
C                  THEMISTOR SETUP
C        RHUM = RELATIVE HUMIDITY
C        RT = THERMISTOR VOLTMETER READING IN MILLIVOLTS
C        TITLE = UP TO 80 CHARACTER TITLE
C        TK = TEMPERATURE IN DEGREES KELVIN
C        XLOC = HYGRISTOR LOCK-IN RESISTANCE
C....  READOUT NOT CORRECTED FOR NONLINEARITY OF THE INSTRUMENT
       CHARACTER*80 TITLE
       CHARACTER*12 FNAME, FILEOUT
       CHARACTER*8 LOT, DAT
       REAL*4 RT,HT,GUNPRES,REFPRES,RVALUE2,RESISRT,RESISHT
      1,ELEVG, ELEVR, AIRSPEED, TK, RHUM, EVAPOR, NEWELEV
      2,MEANT, MPRES, MVAPOR, GUNTEMP, REFTEMP
       REAL*8 XLOC,A,B,C,D, GRAV, GRATIO, PRES, NMINUS1, READOUT
      1,DISTANCE, AMEAN, STDEV
       COMMON /FLYDATA/  TITLE(1), GUNPRES(1), REFPRES(1), RT(100),
      1HT(100), ELEVG(1), ELEVR(1), AIRSPEED(1), NUMBER(1),
      2GUNTEMP(1), REFTEMP(1), READOUT(1)
       COMMON /DATAHYG/ LOT(1), XLOC(1), DAT(1), A(2,1), B(1),
      1C(4,2,1), D(5,2,1)
       DIMENSION RESISRT(50), RESISHT(50), TK(50), RHUM(50), EVAPOR(50)
      1,PRES(50), NMINUS1(50)
       LUN = 3
       WRITE (6,*) 'TYPE FLIGHT DATA FILE NAME **********-ENCLOSE WITH
      & SINGLE QUOTES'
       READ (5,*) FNAME
       WRITE (6,*) 'TYPE OUTPUT FILE NAME'
       READ (5,*) FILEOUT
       OPEN (UNIT = 2 , FILE = 'HYGRISTOR.DAT', STATUS = 'OLD')
       OPEN (UNIT = LUN, FILE = FNAME, STATUS = 'OLD', DEFAULTFILE=
```

```
     &'[].DAT')
      OPEN (UNIT = 7, FILE = FILEOUT, STATUS = 'NEW', DEFAULTFILE=
     &'[].LIS')
C
C.....READ FIGHT DATA AND STORE IN ARAYS
C
      CALL GETFLY ( LUN, IEOF)
      NHYG = 1
C
C.....READ COEFFICIENTS FOR HYGRISTOR EQUATION
C
      CALL RDHYG83 (NHYG)
C
C.....PRINT FLIGHT LINE HEADER CARD AND TITLE CARD INFORMATION
C
      WRITE ( 7, 200) TITLE
 200  FORMAT (1H ,'FLIGHT LINE TITLE CARD  ', A80)
      WRITE (7, 201) GUNPRES, REFPRES
 201  FORMAT (1H ,' GUN PRESSURE = ' ,F6.2, ' MILLIBARS'/
     &1H ,' REFLECTOR PRESSURE = ', F6.2, ' MILLIBARS')
      WRITE (7, 2011) ELEVG, ELEVR, GUNTEMP, REFTEMP,AIRSPEED,
     1READOUT
2011  FORMAT(1H , ' GUN ELEVATION = ', F8.1, ' METERS'/
     11H , ' REFLECTOR ELEVATION = ', F8.1, ' METERS'/
     21H , ' GUN TEMPERATURE = ', F7.2, ' DEGREES CELSIUS'/
     31H , ' REFLECTOR TEMPERATURE = ', F7.2, ' DEGREES CELSIUS'/
     41H , '     AIRCRAFT SPEED = ', F8.1, ' MPH'/
     51H , ' DISTANCE READOUT = ', F12.4, ' METERS')
      WRITE (7, 198)NUMBER
 198  FORMAT(1H , ' NUMBER OF READING =  ', I4)
      WRITE (7, 199) IEOF
 199  FORMAT('IEOF =    ', I4)
C
C.....PRINT HYGRISTOR EQUATION DATA AND COEFFICIENTS
C
      WRITE (7, 2013) LOT, DAT, XLOC
 2013 FORMAT(1H , ' HYGRISTOR EQUATION DATA '/
     11H ,'  LOT NUMBER = ', A8, '  DATE = ', A8, ' LOCK-IN RESISTANCE
     2  ', E10.4)
      WRITE (7, 2015) A, B
 2015 FORMAT(1H ,'  A1 = ', F10.4, '  A2 = ', F10.4,  ' B =  ', F6.1)
      WRITE (7, 2017) C, D
 2017 FORMAT(1H ,'  C AND D COEFFICIENTS =    '/
     11H , 8(E11.4,1X))
C
C.....PRINT RAW VOLTMETER READINGS FOR TRANSDUCERS
C
      WRITE (7, 202) RT
 202  FORMAT(1H ,'THERMISTOR READING IN MILLIVOLTS',/10(F7.3))
      WRITE (7, 203) HT
```

```
 203  FORMAT(1H ,'HYGRISTOR READING IN MILLIVOLTS',/10(F7.3))
C
C.....CONVERT VOLTAGE READINGS TO RESISTANCE
C
      DO 100 I=1, NUMBER(1)
      RESISRT(I) = RVALUE2(RT(I)) * 2.0
      RESISHT(I) = RVALUE2(HT(I))
100   CONTINUE
      WRITE (7, 204) RESISRT
 204  FORMAT(1H ,'THERMISTOR RESISTANCE IN OHMS',/10(F10.2,1X))
      WRITE (7, 205) RESISHT
 205  FORMAT(1H ,'HYGRISTOR RESISTANCE IN OHMS',/10(F10.2,1X))
C
C.....CALCULATE THERMISTOR TEMPERATURES AND PRINT RESULTS
C
      DO 110 I = 1, NUMBER(1)
        IF(RESISRT(I).GT.0.0) THEN
          TK(I) = RAWTEMP (RESISRT(I))
        ELSE
           TK(I) = -999.0
        ENDIF
 110  CONTINUE
       WRITE (7, 206) TK
 206  FORMAT(1H ,'RAW TEMPERATURE IN DEGREES KELVIN',/10(F8.2,1X))
C
C.....CALCULATE RELATIVE HUMIDITY
C
       K = 1
      DO 209 I=1, NUMBER(1)
C
C..... AND CORRECT TEMPERATURE FOR AIRSPEED
C
      TK(I) = TEMPOK (GUNPRES(1), TK(I), AIRSPEED)
      CALL RH33NEW(TK(I), RESISHT(I), K, RHUM(I))
 209  CONTINUE
      WRITE (7, 211) RHUM
 211  FORMAT(1H ,'  RELATIVE HUMIDITY'/10(F8.3))
C
C.....CALCULATE VAPOR PRESSURE
C
      DO 213 I=1, NUMBER(1)
      CALL VAPORP (RHUM(I), TK(I), EVAPOR(I))
 213  CONTINUE
      WRITE (7, 215) EVAPOR
 215  FORMAT(1H ,'   VAPOR PRESSURE IN MILLIBARS '/
     210(F8.3, 1X))
C
C.....GO INTO DO LOOP TO CALCULATE PRESSURE FOR EACH MEASURED POINT
C.....ASSUME CONSTANT AIRCRAFT VELOCITY
C
```

```
      DO  300 J = 1, NUMBER(1)
      ELEVDIFF = ELEVG(1) - ELEVR(1)
      ELEVINT = ELEVDIFF / FLOAT (NUMBER(1) - 1)
      NEWELEV =  (ELEVINT * FLOAT(J)) - ELEVINT
C
C.....CALCULATE MEAN PRESSURE, TEMPERATURE, VAPOR PRESSURE
C
      IF(J.EQ.1) THEN
         MEANT = TK(J)
         MPRES = GUNPRES(1)
         MVAPOR = EVAPOR(1)
      ELSEIF(J.EQ.2) THEN
         MEANT = (TK(J - 1) + TK(J)) / 2.0
         MPRES = GUNPRES(1)
         MVAPOR = (EVAPOR(J - 1) + EVAPOR(J)) / 2.0
      ELSE
         MEANT = (TK(J - 1) + TK(J)) / 2.0
         MPRES = (PRES(J - 2) + PRES(J - 1)) / 2.0
         MVAPOR = (EVAPOR(J - 1) + EVAPOR(J)) / 2.0
      ENDIF
C
C.....CALCULATE GRAVITY ACCELERATION RATIO FOR PRESSURE EQUATION
C
      GELEV = ABS(NEWELEV)
      GRAV = GRATIO (GELEV)
      CALL APRESS (J, NEWELEV, MEANT, MPRES, MVAPOR, GRAV, PRES(J))
      CALL HPINDEX ( PRES(J), TK(J), EVAPOR(J), NMINUS1(J))
 300  CONTINUE
C
C.....WRITE OUT INDEX ARRAY
C
      WRITE (7, 310) NMINUS1
 310  FORMAT(1H ,' INDEX OF REFRACTION MINUS 1 = '/
     110(E12.7,1X))
C
C.....CALCULATE MEAN REFRACTIVE INDEX AND STANDARD DEVIATION
C
      CALL MEAND(NUMBER(1), NMINUS1, STDEV, AMEAN)
      WRITE (7,312) AMEAN, STDEV
 312  FORMAT(1H ,' MEAN REFRACTIVE INDEX = ', E12.5,/
     11H ,' STANDARD DEVIATION = ', E12.5)
      AMEAN = AMEAN + 1.0
C
C.....COMPUTE DISTANCE USING STANDARD REFRACTIVE INDICES
C.....THE FIRST FOR 0 DEGREE C, 1013.25 MILLIBAR PRESSURE
C.....EQ. FROM BOMFORD
C.....THE SECOND FOR 15 DEGREE C, 760MMHG, FROM HP
C.....BOMFORD = 1.00028994, DISTANCE 1
C.....HP = 1.00027926, DISTANCE 2
      DISTAN1 = READOUT(1) * (1.00028994 / AMEAN)
```

```
      DISTAN2 = READOUT(1) * (1.00027926 / AMEAN)
C
C....CALCULATE MEAN TEMPERATURE ALONG LINE
C
      CALL MEANS(NUMBER(1), TK, STDEV, AMEAN)
      WRITE (7,314) AMEAN, STDEV
 314  FORMAT(1H ,' MEAN TEMPERATURE = ', F8.2, ' DEGREES KELVIN'/
     11H ,' STANDARD DEVIATION = ', F8.3)
      WRITE(7, 316) DISTAN1
 316  FORMAT(1H ,' CORRECTED DISTANCE-1 = ', F12.4, ' METERS',
     1' , USING BOMFORD FORMULA')
      WRITE(7, 318) DISTAN2
 318  FORMAT(1H ,' CORRECTED DISTANCE-2 = ', F12.4,  ' METERS',
     1' , USING HP FORMULA')
      CLOSE (7)
      STOP
      END
```

COMMON.FOR

```
C.....COMMON.FOR... INCLUDE FILE FOR AVEINDEX.FOR SUBROUTINES
      CHARACTER*80 TITLE
      CHARACTER*8 LOT, DAT
      REAL*4 RT, HT, GUNPRES, REFPRES, RVALUE2, RESISTRT, RESISHT
     1,ELEVG, ELEVR, AIRSPEED, TK, RHUM, EVAPOR, NEWELEV
     2,MEANT,MPRES, MVAPOR, GUNTEMP, REFTEMP
      REAL*8 XLOC, A, B, C, D, GRAV, PRES
      COMMON /FLYDATA/ TITLE(1), GUNPRES(1), REFPRES(1), RT(100),
     1HT(100), ELEVG(1), ELEVR(1), AIRSPEED(1), NUMBER(1),
     1GUNTEMP(1), REFTEMP(1), READOUT(1)
      COMMON /DATAHYG/ LOT(1), XLOC(1), DAT(1), A(2,1), B(1)
     1,C(4,2,1), D(5,2,1)
```

FLYENTER.FOR

```
C.....PROGRAM FLYENTER.FOR IS A DATA ENTRY PROGRAM FOR CREATING
C.....DATA FILES FROM FLIGHT DATA COLLECTED DURING EDM OF CVO
C.....LINES. THE PROGRAM FIRST COLLECTS HEADER CARD INFORMATION,
C.....TITLE CARD INFO AND PAIRS OF VOLTAGE MEASUREMENTS FROM THE
C.....AIRCRAFT RECORDER FOR THE THERMISTOR AND HYGRISTOR
C.....ELECTRONICS OUTPUT.
C..... VARIABLES:
C               GUNPRES = GUN PRESSURE IN MILLIBARS
C               REFPRES = REFLECTOR PRESSURE IN MILLIBARS
C               ELEVG = GUN ELEVATION IN METERS
C               ELEVR = REFLECTOR ELEVATION IN METERS
C               GUNTEMP = GUN TEMPERATURE IN DEGREES CELSIUS
C               REFTEMP = REFLECTOR TEMPERATURE IN DEGREES CELSIUS
C               AIRSPEED = AIRCRAFT VELOCITY IN MILES PER HOUR
C                          (NOTE, PROGRAM DOES NOT CONVERT FROM
C                           KNOTS TO MPH)
C               READOUT = DISTANCE READOUT FROM RANGE-MASTER
C               RT = THERMISTOR READING IN MILLIVOLTS
C               HT = HYGRISTOR READING IN MILLIVOLTS
C....PROGRAM CHECKS FOR OUT OF RANGE RT AND HT VALUES
C....EXAMPLE INPUT:
C     'LINE1'                    !FILE NAME, DEFAULT FILE TYPE = .DAT
C      794.46 2058.80 10.56      !GUN DATA
C      720.89 2822.8 5.0         !REFLECTOR DATA
C      35. 11111.111             !AIRSPEED, RANGE MASTER READOUT
C     'R8 - 6718, FLIGHT SPEED = 35 MPH, READING IN MV EVERY 5 SEC.'
C     8.21 -.28                  !PAIRS OF THERMISTOR, HYGRISTOR READINGS
C     8.19 -.27
C     8.00 -.26
C     9999 9999                  !TO END ENTRY PROGRAM
C
C
      CHARACTER*80 TITLE
      CHARACTER*12 FNAME
      REAL*4 GUNPRES,REFPRES,ELEVG,ELEVR,GUNTEMP,REFTEMP,AIRSPEED
      REAL*8 READOUT
      WRITE (6,*) 'TYPE FILE NAME ******** ENCLOSE WITH SINGLE QUOTES'
      READ (5,*) FNAME
      OPEN (UNIT = 2, FILE = FNAME, STATUS = 'NEW',
     1DEFAULTFILE = '[].DAT')
      WRITE(6,*) 'TYPE GUN PRESSURE, GUN ELEVATION, GUN TEMP'
      WRITE(6,*) 'SEPARATE BY BLANK'
      WRITE(6,*) 'TYPE 0 IF NO DATA FOR GUN TEMP, OTHERS REQUIRED'
       READ(5,*)GUNPRES,ELEVG,GUNTEMP
      WRITE(6,*) 'TYPE REFLECTOR PRESSURE, ELEVATION, TEMP'
      WRITE(6,*) 'ELEVATION REQUIRED, OTHERS FOR INTERNAL CHECKS'
      READ(5,*)REFPRES,ELEVR,REFTEMP
      WRITE(6,*) 'TYPE AIRSPEED IN MPH AND RANGEMASTER READOUT'
```

```
      WRITE(6,*) 'AIRSPEED REQUIRED'
      READ(5,*) AIRSPEED, READOUT
      WRITE(2,100)GUNPRES,ELEVG,GUNTEMP,REFPRES,ELEVR,REFTEMP
     1,AIRSPEED,READOUT
  100 FORMAT(1H ,'HEADER:',2(F8.2,F8.1,F7.2),F8.1,F14.4)
      WRITE(6,*) 'TYPE TITLE, UP TO 80 CHARACTERS ENCLOSE WITH SINGLE
     1QUOTES'
      READ(5,*) TITLE
      WRITE(2,110) TITLE
  110 FORMAT(1H ,'TITLE:', A80)
      WRITE(6,*) 'ENTER DATA PAIRS, SEPARATE BY SPACE'
      WRITE(6,*) 'TYPE 9999 9999 TO END ENTRY'
  10  READ(5,*)RT, HT
      IF(RT.GE.9999.) GO TO 800
      IF(HT.GE.9999.) GO TO 800
C
C.....TEST FOR BAD INPUT DATA
C
      IF(RT.GE.100.0.OR.RT.LE.-100.0)THEN
         WRITE(6,*)'POSSIBLE ERROR IN RECORDED DATA'
         WRITE(6,*)'RT ***** OUT OF RANGE'
      ELSEIF(HT.GE.100.0.OR.HT.LE.-100.0) THEN
         WRITE(6,*)'POSSIBLE ERROR IN RECORDED DATA'
         WRITE(6,*)'HT ***** OUT OF RANGE'
      ENDIF
      WRITE(2,112) RT, HT
  112 FORMAT(1H ,2F7.2)
      GO TO 10
  800 CONTINUE
      WRITE(6,*)'END ENTRY'
      CLOSE (2)
      STOP
      END
```

GETFLY.FOR

```
      SUBROUTINE GETFLY( LUN, IEOF)
C.....SUBROUTINE TO READ FLIGHT DATA FROM EDM LINES
C.....APPROPRIATE UNIT (FILE) TO BE OPENED IN MAIN PROGRAM, THIS VERSION
C.....FOR THE CVO VAX SYSTEM AND CVO FLIGHT DATA BY E. ENDO 1983
C..... FORMAL PARAMETERS:
C.....  LUN = UNIT NUMBER TO BE READ FROM
C.....  IEOF = FLAG FOR END-OF-FILE (=0 IF NORMAL;=1 IF IEOF ENCOUNTERED)
C.....  FIRSCAR = FIRST CHARACTER ON DATA CARDS USED AS FLAG
C.....  GUNPRES = GUN AIR PRESSURE IN MILLIBARS
C.....  REFPRES = REFLECTOR AIR PRESSURE IN MILLIBARS
C.....  ELEVG, ELEVR = ELEVATION OF GUN AND REFLECTOR IN METERS
C.....  GUNTEMP, REFTEMP = TEMPERATURE IN DEGREES CELSIUS OF END POINTS
C.....  AIRSPEED = AIRCRAFT SPEED IN MILES PER HOUR (NOT KNOTS)
C.....  READOUT = DISTANCE READOUT OF RANGE MASTER
C.....  RT = THERMISTOR RAW DIGITAL VOLTMETER READINGS
C.....  HT = HYGRISTOR RAW DIGITAL VOLTMETER READINGS
      CHARACTER*80 CARD, TITLE
      CHARACTER*1 H,T,BLNK, FIRSCAR
      REAL*4 GUNPRES, REFPRES, RT, HT, ELEVG, ELEVR, AIRSPEED
     1,GUNTEMP, REFTEMP
      COMMON /FLYDATA/ TITLE(1),GUNPRES(1),REFPRES(1),RT(100),HT(100)
     1,ELEVG(1), ELEVR(1), AIRSPEED(1),NUMBER(1), GUNTEMP(1), REFTEMP(1)
     2,READOUT(1)
      DATA H, T, BLNK /'H', 'T', ' '/
      IEOF = 0
      NUMBER(1) = 0
      N = 0
 100  READ ( LUN, '(A80)' , END = 900, ERR = 800) CARD
      N = N + 1
      READ (CARD(2:2), '(A1)', END = 900, ERR = 800) FIRSCAR
        IF(FIRSCAR.EQ.H) THEN
          READ (CARD(9:31), '(F8.2,F8.1,F7.2)', END = 900, ERR = 800)
     &GUNPRES, ELEVG, GUNTEMP
          READ (CARD(32:54), '(F8.2,F8.1,F7.2)', END = 900, ERR =800)
     1REFPRES, ELEVR, REFTEMP
          READ (CARD(55:62), '(F8.1)', END =900, ERR = 800) AIRSPEED
          READ (CARD(63:74), '(F12.4)', END = 900, ERR = 800) READOUT
        ELSEIF(FIRSCAR.EQ.T) THEN
          READ (CARD(2:80), '(A80)', END = 900, ERR = 800)TITLE
        ELSEIF(FIRSCAR.EQ.BLNK) THEN
          NUMBER(1) = NUMBER(1) + 1
          READ (CARD(2:15), '(2(F7.2))', END = 900, ERR = 800)
     &RT(NUMBER(1)),HT(NUMBER(1))
        ENDIF
      GO TO 100
 800  IEOF = 1
 900  CONTINUE
      WRITE(7, 999) N
```

```
999  FORMAT(1H ,'  NUMBER OF DATA CARDS READ =    ', I4)
     RETURN
     END
```

GRATIO.FOR

```
      FUNCTION GRATIO(ELEV)
C.....FUNCTION GRATIO RETURNS THE RATIO OF G45 TO THE ACCELERATION
C.....OF GRAVITY CALCULATED FOR A GIVEN ELEVATION.
C.....G AT 45 DEGREES LATITUDE IS THE REFERENCE VALUE
C.....FOR THE ACCELERATION OF GRAVITY AT SEA LEVEL
C..... 0.0003086 CM/SEC**2/M IS SUBTRACTED FROM THIS BASE VALUE
C..... VARIABLES USED:
C..... G45 = ACCELERATION OF GRAVITY AT 45 DEGREES LATITUDE
C.....     = 980.621 CM/SEC**2, FROM CRC HANDBOOK
C..... BASEG = ACCELERATION OF GRAVITY AT GUN ELEVATION
C..... ELEV = ELEVATION USED FOR CALCULATION IN METERS,
C.....        INITIALLY CALCULATED ELEVATION CHANGE
C..... ELEVG = GUN ELEVATION IN METERS
C..... CALCULATED ACCELERATION FOR GRAVITY AT GIVEN ELEVATION
      REAL*8 GRAVITY, GRATIO, BASEG
      INCLUDE 'COMMON.FOR'
      ELEV = ELEV + ELEVG(1)
      BASEG = 980.621 - 0.0003086 * ELEVG(1)
      GRAVITY = BASEG - 0.0003086 * ELEV
      GRATIO = BASEG / GRAVITY
      RETURN
      END
```

GRAVITY.FOR

```
      FUNCTION GRATIO(ELEV)
C.....FUNCTION GRATIO RETURNS VALUE FOR THE ACCELERATION OF GRAVITY
C.....FUNCTION GRATIO RETURNS THE RATIO OF G45 TO THE ACCELERATION
C.....OF GRAVITY CALCULATED FOR A GIVEN ELEVATION
C.....AT A GIVEN ELEVATION ABOVE THE EARTH'S SURFACE, SEA LEVEL
C.....G AT 45 DEGREES LATITUDE IS THE REFERENCE VALUE
C..... 0.0003086 CM/SEC**2/M IS SUBTRACTED FROM THIS BASE VALUE
C..... VARIABLES USED:
C..... G45 = ACCELERATION OF GRAVITY AT 45 DEGREES LATITUDE
C.....     = 980.621 CM/SEC**2, FROM CRC HANDBOOK
C..... ELEV = ELEVATION GIVEN IN METERS
      REAL*8 GRAVITY
      GRAVITY = 980.621 - 0.0003086 * ELEV
      GRATIO = 980.621 / GRAVITY
      RETURN
      END
```

HPINDEX.FOR

```
      SUBROUTINE HPINDEX(PRES, TK, EVAPOR, NMINUS1)
C.....INDEX RETURNS THE INDEX OF REFRACTION FOR RED LIGHT
C.....WAVE LENGTH = 0.840 MICROMETERS FOR GALIUM-AS LASER
C..... FOR THIS TEST ROUTINE N0 = 1.000289939706 - 1.0
C..... FOR ALL OTHER INSTRUMENTS, ESPECIALLY K&E RM, N0 HAS TO BE
C..... RECALCULATED USING EQUATION 1.9 FROM BOMFORD, 1980, GEODESY
C..... VARIABLES USED:
C..... NMINUS1 = REFRACTIVE INDEX MINUS 1.0
C..... N0 = REFRACTIVE INDEX FOR STANDARD ATMOSPHERE
C..... PRES = AIR PRESSURE IN MILLIBARS
C..... TK = TEMPERATURE IN DEGREES KELVIN
C..... EVAPOR = WATER VAPOR PRESSURE
      REAL*8 NMINUS1, N0
      N0 = .2899397E-3
      NMINUS1 = (0.2696 * (N0) * (PRES/TK)) -
     1 (11.2 * (EVAPOR / TK) * 1.0E-6)
      RETURN
      END
```

HPINDEX2.FOR

```
      SUBROUTINE HPINDEX2(PRES, TK, EVAPOR, NMINUS1)
C.....INDEX RETURNS THE INDEX OF REFRACTION FOR RED LIGHT
C.....WAVE LENGTH = 0.840 MICROMETERS FOR GALIUM-AS LASER
C..... FOR THIS TEST ROUTINE NO = 1.000294666 - 1.0
C..... FOR ALL OTHER INSTRUMENTS, ESPECIALLY HP3808, NO HAS TO BE
C..... RECALCULATED USING EQUATION 1.9 FROM BOMFORD, 1980, GEODESY
C.....
C.....CONVERTS MILLIBARS TO MMHG
C..... EQN FROM DAVIS, SURVEYING
      REAL*8 NMINUS1, NO
      NO = .294666E-3
      PRES = PRES / 1.33322
      EVAPOR = EVAPOR / 1.33322
      NMINUS1 = (0.359474 * (NO) * (PRES/TK)) -
     1 (1.5026 * (EVAPOR / TK) * 1.0E-5)
      RETURN
      END
```

MEAN.FOR

```
      SUBROUTINE MEAN(NUMBER, VARIABLE, STDEV, AMEAN)
C.....PROGRAM TO CALCULATE THE MEAN AND VARIANCE OF 'N' SAMPLES
C..... ROUTINE FROM DAVIS, STATISTICS AND DATA ANALYSIS IN GEOLOGY
C
C     SET SUMS TO ZERO
      REAL*8 SUMX,SUMX2,VAR,AMEAN,STDEV,VARIABLE
      DIMENSION VARIABLE(50)
      WRITE(7,90)VARIABLE
 90   FORMAT(1H ,10F8.2)
      SUMX = 0.0
      SUMX2 = 0.0
      DO 100 I=1, NUMBER
      SUMX = SUMX + VARIABLE(I)
      SUMX2 = SUMX2 + VARIABLE(I)*VARIABLE(I)
 100  CONTINUE
C
C.... COMPUTE VARIANCE
C
      VAR = (FLOAT(NUMBER)*SUMX2-SUMX*SUMX)/FLOAT(NUMBER*(NUMBER-1))
C
C.... COMPUTE STANDARD DEVIATION, AND MEAN
C
      STDEV = SQRT(VAR)
      AMEAN = SUMX/FLOAT(NUMBER)
      RETURN
      END
```

MEAND.FOR

```
      SUBROUTINE MEAND(NUMBER, VARIABLE, STDEV, AMEAN)
C.....PROGRAM TO CALCULATE THE MEAN AND VARIANCE OF 'N' SAMPLES
C..... ROUTINE FROM DAVIS, STATISTICS AND DATA ANALYSIS IN GEOLOGY
C..... DOUBLE PRECISION VERSION
C
C     SET SUMS TO ZERO
      REAL*8 SUMX,SUMX2,VAR,AMEAN,STDEV,VARIABLE
      DIMENSION VARIABLE(50)
      SUMX = 0.0
      SUMX2 = 0.0
      DO 100 I=1, NUMBER
      SUMX = SUMX + VARIABLE(I)
      SUMX2 = SUMX2 + VARIABLE(I)*VARIABLE(I)
 100  CONTINUE
C
C.... COMPUTE VARIANCE
C
      VAR = (FLOAT(NUMBER)*SUMX2-SUMX*SUMX)/FLOAT(NUMBER*(NUMBER-1))
C
C.... COMPUTE STANDARD DEVIATION, AND MEAN
C
      STDEV = SQRT(VAR)
      AMEAN = SUMX/FLOAT(NUMBER)
      RETURN
      END
```

MEANS.FOR

```
      SUBROUTINE MEANS(NUMBER, VARIABLE, STDEV, AMEAN)
C.....PROGRAM TO CALCULATE THE MEAN AND VARIANCE OF 'N' SAMPLES
C..... ROUTINE FROM DAVIS, STATISTICS AND DATA ANALYSIS IN GEOLOGY
C
C     SET SUMS TO ZERO
      REAL*4 SUMX,SUMX2,VAR,AMEAN,STDEV,VARIABLE
      DIMENSION VARIABLE(50)
      WRITE(7,90)VARIABLE
 90   FORMAT(1H ,10F8.2)
      SUMX = 0.0
      SUMX2 = 0.0
      DO 100 I=1, NUMBER
      SUMX = SUMX + VARIABLE(I)
      SUMX2 = SUMX2 + VARIABLE(I)*VARIABLE(I)
 100  CONTINUE
C
C.... COMPUTE VARIANCE
C
      VAR = (FLOAT(NUMBER)*SUMX2-SUMX*SUMX)/FLOAT(NUMBER*(NUMBER-1))
C
C.... COMPUTE STANDARD DEVIATION, AND MEAN
C
      STDEV = SQRT(VAR)
      AMEAN = SUMX/FLOAT(NUMBER)
      RETURN
      END
```

MVOLTTOR.FOR

```
C.....FUNCTION TO RETURN RESISTANCE GIVEN READING IN MILLIVOLTS FOR
C.....DATEL PRINTER AND ASSOCIATED ELECTRONICS - E. ENDO 1983
      FUNCTION RVALUE2(MVOLTS)
C.....LINEAR RELATION DETERMINED FROM CALIBRATION DATA
C..... VARIABLES:
C.....  RVALUE2 = RESISTANCE FOR TRANSDUCER IN OHMS
      REAL*4 RVALUE2
      RVALUE2 = 15184.37 - 276.61419 * (MVOLTS)
      RETURN
      END
```

RAWTEMP.FOR

```
      FUNCTION RAWTEMP(RT)
C.....FUNCTION TO RETURN RAW TEMPERATURE IN DEGREES KELVIN GIVEN
C.....THEMISTOR RESISTANCE MEASURED, FUNCTION IS SPECIFIC TO VIZ
C.....THEMISTOR LOT NUMBER P6970, #1366-205 PREMIUM
C.....FUNCTION USES 3RD ORDER POLYNOMIAL EQUATION AND COEFFICIENTS
C.....SUPPLIED BY VIZ MANUFACTURING
C..... VARIABLES USED:
C.....           RT = RESISTANCE MEASURED
C.....           R30 = LOCK=IN RESISTANCE, NOMINAL = 14000 OHMS
C.....           RAWTEMP = UNCORRECTED TEMPERATURE IN DEGREES KELVIN
      REAL*8 RAWTEMP, RT, RSRATIO
C
      R30 = 14000.0
      RSRATIO = DLOG(RT / R30)
       RAWTEMP = 1.0 / (3.2987D-3 + (4.7610D-4 * RSRATIO) + (2.8417D-6 *
     &(RSRATIO * RSRATIO)) +(1.5691D-6 * (RSRATIO * RSRATIO * RSRATIO)))
      RETURN
      END
```

RDHYG83.FOR

```
      SUBROUTINE RDHYG83 ( NHYG )
C.....SUBROUTINE TO READ HYGRISTOR CONSTANTS FROM A DATA FILE
C.....FROM PRESCOTT - MODIFIED BY E. ENDO 7-6-83, CVO
C.....TO READ CONSTANTS IN E FORMAT FOR PRECISION
C.....REQUIRED IN EQUATION 2 CONSTANTS
C.....LOT = LOT NUMBER
C....A = A COEFFICIENTS FROM VIZ WRITEUP
C....B = B COEFFICIENTS FROM VIZ WRITEUP
C....C = C COEFFICIENTS FROM VIZ WRITEUP
D....D = D COEFFICIENTS FROM VIZ WRITEUP
C....XLOC = LOCK-IN RESISTANCE IN OHMS
      REAL*8 A,B,C,D, XLOC
      COMMON /DATAHYG/ LOT(1),XLOC(1),DAT(1),A(2,1),B(1),C(4,2,1),
     &D(5,2,1)
      CHARACTER*8 LOT, DAT, BLANK
      K = 1
      READ (2,800, ERR = 900, END= 900) LOT(K), A(1,K), A(2,K), B(K),
     &XLOC(K), DAT(K)
 800  FORMAT(A8,2X,4(F10.0),A8)
      IF (LOT(K).EQ.BLANK) RETURN
C.....READ D AND C CONSTANTS FOR EQUATION, E FORMAT IDENTICAL TO VIZ WRITEUP
      READ(2,801, ERR = 900, END= 900) ((D(I, J, K), I =1, 5), J=1,2),
     &((C(I, J, K),I = 1,4),J=1,2)
 801  FORMAT(5(E11.4,1X)/5(E11.4,1X)/4(E11.4,1X)/4(E11.4,1X))
      NHYG = K
 900  CONTINUE
      RETURN
      END
```

RH33NEW.FOR

```
      SUBROUTINE RH33NEW(TEMP, HDRG, K, RHUM)
C.....SUBROUTINE RH33NEW TO CALCULATE HUMIDITY - PERCENT DIVIDED BY 100.0
C.....MODIFIED 7-7-83 BY E.ENDO
C.....TEMP = TEMPERATURE IN DEGREES KELVIN
C.....TEMPC = TEMPERATURE IN DEGREES CELSIUS
C.....HDRG = RESISTANCE READING FROM HYGRISTOR
C.....IN ORIGINAL FUNCTION HDRG IS THE READING R / LOC-IN R RATIO
C.....A, B, C, D, = POLYNOMIAL COEFFICIENTS FOR EQUATION, SELECTION FROM 4
C.....POSSIBLE CHOICES DEPEND ON HRDG / XLOC RATIO AND RELATIVE HUMIDITY
C.....IF INITIAL CALCULATION SHOWS LESS THAN 20%, CALCULATE USING A
C..... ANOTHER SET OF COEFFICIENTS
      REAL*8 A, B, C, D, RATIO, SUMT, FRS, SUMD, RH33
     1, XLOC
      REAL*4 HDRG, TEMP, RHUM
      CHARACTER*8 LOT, DAT
      COMMON /DATAHYG/ LOT(1), XLOC(1), DAT(1), A(2,1), B(1),
     2C(4,2,1), D(5,2,1)
C
C.....COMPUTE HDRG / LOCK-IN RESISTANCE RATIO TO DETERMINE EQUATION 2
C.....COEFFICIENTS
        RATIO = HDRG / XLOC(1)
        IF(RATIO.LT.1.) THEN
          J = 2
         ELSE
          J = 1
         ENDIF
C
C.....CONVERT DEGREES KELVIN TO DEGREES CELSIUS
C
      TEMPC = TEMP - 273.15
      SUMT = C(1, J, K)
      DO 2 I = 1, 3
         SUMT = SUMT + C(I + 1, J, K) * (TEMPC **I)
 2    CONTINUE
C
      FRS = SUMT * DLOG(RATIO)
C.....TEST FRS TO DETERMINE POLYNOMIAL COEFFICIENTS FOR EQUATION 1 CONSTANTS
      IF (FRS.LT.-0.2) THEN
         J = 2
      ELSE
         J = 1
      ENDIF
C
      SUMD = D(1, J, K)
      DO 4 I=1, 4
      SUMD = SUMD + D(I+1, J, K) * ( FRS ** I)
 4    CONTINUE
      RH33 = A(J, K) - (B(K)/SUMD)
```

```
RHUM = RH33 /100.0
RETURN
END
```

RINDEX.FOR

```
      SUBROUTINE RINDEX(PRES, TK, EVAPOR, NMINUS1)
C.....INDEX RETURNS THE INDEX OF REFRACTION FOR RED LIGHT
C.....WAVE LENGTH = 0.6328 MICROMETERS FOR HELIUM-NEON LASER
C..... FOR THIS TEST ROUTINE NO = 1.000291754 - 1.0
C..... FOR ALLL OTHER INSTRUMENTS, ESPECIALLY HP3808, NO HAS TO BE
C..... RECALCULATED USING EQUATION 1.9 FROM BOMFORD, 1980, GEODESY
C..... NO
      REAL*8 NMINUS1, NO
      NO = .291754E-3
      NMINUS1 = (0.2696 * (NO) * (PRES/TK)) -
     1 (11.2 * (EVAPOR / TK) * 1.0E-6)
      RETURN
      END
```

RVALUE2.FOR

```
      FUNCTION RVALUE2(MVOLTS)
C.....FUNCTION TO RETURN RESISTANCE GIVEN READING IN MILLIVOLTS FOR
C.....DATEL PRINTER AND ASSOCIATED ELECTRONICS - E. ENDO 1983
C.....LINEAR RELATION DETERMINED FROM CALIBRATION DATA
C.....SPECIFIC TO CVO ELECTRONICS WHICH IS IDENTICAL TO THAT
C.....USED BY SAVAGE, ET. AL. VARIATION A RESULT OF SMALL DIFFERENCES
C.....IN COMPONENT VALUES USED IN THE CIRCUITRY
C..... VARIABLES:
C.....  RVALUE2 = RESISTANCE OF TRANSDUCER IN OHMS
      REAL*4 RVALUE2,MVOLTS
      RVALUE2 = 15184.37 - (276.61419 * MVOLTS)
      RETURN
      END
```

TEMPOK.FOR

```
      FUNCTION TEMPOK(GUNPRESS, RAWTEMP, AIRSPD)
C.....FUNCTION TO RETURN TEMPERATURE CORRECTED FOR AIRSPEED, AIRSPEED
C.....CORRECTED FOR APPROXIMATE AIR PRESSURE ALONG LINE. PROGRAM
C.....USES INSTRUMENT PRESSURE FOR THIS CORRECTION
C.....VARIABLES USED:
C.....   TASK = RECORDED AIRSPEED IN KNOTS
C.....   TASM = RECORDED AIRSPEED IN STATUE MILES PER HOUR
C.....   TRUEASP = AIRSPEED CORRECTED FOR PRESSURE
C.....   GUNPRESS = AIR PRESSURE AT INSTRUMENT IN MILLIBARS
C.....   RAWTEMP  = RAW THERMISTOR TEMPERATURE
C.....   AIRSPD  = AIRSPEED IN MPH
C.....   TEMPOK = CORRECTED TEMPERATURE IN DEGREES KELVIN
C.....      EQUATIONS FROM REFR33.FOR, PRESCOTT, ET AL. AND
C.....   SAVAGE, 1973
C..... MODIFY FOR DOUBLE PRECISION IF NEEDED
C..... PRESSURE CORRECTION FOR TRUE AIRSPEED IS AN APPROXIMATION SINCE
C..... RAWTEMP HAS TO BE USED RATHER THAN CORRECTED TEMPERATURE
      TCONS = 16667.0
C
C.....CONVERT MILLIBARS TO mmHG TO RETAIN CHARACTER OF ORIGINAL EXPRESSION
C
      GPRESS = GUNPRESS / 1.33322
      TRUEASP = AIRSPD * SQRT(760./GPRESS * RAWTEMP/288.)
      TEMPOK = RAWTEMP - TRUEASP * TRUEASP / TCONS
      RETURN
      END
```

VAPOR.FOR

```
      SUBROUTINE VAPORP(R, TK, EVAPOR)
C.....SUBROUTINE VAPOR.FOR
C.....CALCULATES VAPOR PRESSURE GIVEN RELATIVE HUMIDITY IN PERCENT, R
C.....USES EQUATION 2.89 FROM FLEAGLE AND BUSINGER, 1963, AN INTRODUCTION
C.....TO ATMOSPHERIC PHYSICS
C.....SATURATION VAPOR PRESSIRE GIVEN AS
C....  ES = 10** (9.4051 - 2353 / TK), TK = TEMPERATURE IN DEGREES KELVIN
C....     ES IS RESULT IN MILLIBARS, ABOUT 4% ERROR AT 30 DEG C AND
C...   2% ERROR AT 20 DEGREES CELSIUS
C...   VAPOR PRESSURE CALCULATED FROM RELATIVE HUMIDITY, USE DEFINITION
C...      R = E / ES, OR E = R * ES
C.....VARIABLES USED
C.....  TK = MEASURED TEMPERATURE, KELVIN
C.....  EVAPOR = VAPOR PRESSURE, IN MILLIBARS
C.....  ES = SATURATION VAPOR PRESSURE, IN MILLIBARS
C.....  R = RELATIVE HUMIDITY IN PERCENT
      REAL*4 R,ES,EVAPOR, TK
      ES = 10.0 ** (9.4051 - 2353. / TK)
      EVAPOR = ES * R
      RETURN
      END
```

VOLTSTOR.FOR

```
C.....FUNCTION TO RETURN RESISTANCE GIVEN READING IN VOLTS FOR
C.....DATEL PRINTER AND ASSOCIATED ELECTRONICS - E. ENDO 1983
      FUNCTION RVALUE(VOLTS)
C.....LINEAR RELATION DETERMINED FROM CALIBRATION DATA
C..... VARIABLES:
C.....  RVALUE = RESISTANCE FOR TRANSDUCER IN OHMS
      RVALUE = 15184.37 - 276614.19* (VOLTS)
      RETURN
      END
```

References Cited

Berg, E., J. A. Carter, D. Harris, S. H. Laurila, B. E. Schenck, G. H. Sutton, J. E. Wolfe, and S. F. Cushman, High-Precision laser distance measurement in support of Lunar laser ranging at Haleakala, Maui, 1976-1977, Hawaii Institute Geophysics report HIG-78-5, 184 pp., 1978.

Bomford, G., Geodesy, 4th edition, Clarendon Press, London, 855 pp., 1980.

Fleagle, R. G., and J. A. Businger, An introduction to atmospheric physics, Academic Press, New York, 346 pp., 1963.

Savage, J. C., and W. H. Prescott, Precision of geodolite distance measurements for determining fault movements, J. Geophys. Res. 78, 6001-6008, 1973.

West, R. C., editor, Handbook of Chemistry and Physics, The Chemical Rubber Company, 1969.

Figure 1. Calibration data for converting voltage readings from transducer electronics to resistance in ohms.